A Brief Tour of SAS for Windows
SAS is one of the most versatile and comprehensive statistical software packages available today, with data management, analysis, and graphical capabilities. It is great for working with large databases. SAS has many idiosyncrasies and carry-overs from its initial development in a mainframe environment. This document will introduce some of the key concepts for working with data using SAS software in a Windows environment.  

The SAS Desktop

When you open SAS, you will see the SAS desktop with three main windows:

1. The Editor window

This is the window where you create, edit, and submit SAS command files. The default editor is the Enhanced Editor, which has a system of color coding to make it easier to edit and trouble-shoot command files.

2. The Log window

This is the window where SAS will echo all of your commands, along with any notes (shown in blue), error messages (shown in red), and warnings (shown in green). The log window is cumulative throughout your session and helps to locate any possible problems with a SAS program.

3. The Explorer window

Among other things, this window shows the libraries that you have defined. SAS libraries are folders that can contain SAS datasets and catalogs (e.g., formats catalogs). When you start SAS, you will automatically have the libraries WORK, SASUSER, and SASHELP defined, plus Maps, if you have SAS/Maps on your system. You can define other libraries where you can store and access datasets, as we will see later. If you accidentally close this window, go to View > Contents Only, to reopen it. 

Additional Windows
1. The Output window
This window will be behind other windows until you generate some output.  The text in this window can be copied and pasted to a word processing program, but it cannot be edited or modified. When copying text from the output window, be sure your selection stops on a line that has some text on it. If you select an area that is after the end of the text, you will get an error.

2. The SAS/Graph window

This window will not open until you generate graphs using procedures such as Proc Gplot or Proc Univariate.

Navigating Windows
You can navigate among the SAS workspace windows by clicking on the navigation bar at the bottom, or by using the Window menu to select the desired window. If you have closed a window, you can re-open it by going to the View menu. Menu options available in each window are context-sensitive, and vary depending on the window. 

Set the current directory

Setting the current directory specifies a default location when reading and writing raw data files, graphical output files, and SAS command files. At the beginning of your SAS session, double-click on the directory location at the bottom of the SAS workspace window. You will be able to browse to the folder to use. 

Browse to the folder you want to use,, and then click “OK”. Remember, you are choosing a folder location, and not a specific file. 

SAS Font Setup
If you generate SAS output and then print it on a computer that has SAS installed on it, you won’t have any problems with the SAS output looking weird or strange, but if you try to print your SAS output on a computer that doesn’t have SAS installed on it, you may get some very odd-looking output. To make sure you have nice-looking output every time you use SAS, include and submit the following line of code as the first thing you do each time you invoke SAS. 

OPTIONS FORMCHAR="|----|+|---+=|-/\<>*";

You can get this bit of code from my web page at http://www.umich.edu/~kwelch/fonts.sas
 SAS Help
When you first open SAS, you will have the option to open SAS help by clicking on "Start Guides" in the Getting Started with SAS window.

If you close this window you can start SAS help later by going to Help > SAS Help and Documentation.

To get help on statistical procedures, click on the Contents tab > SAS Products > SAS/Stat > SAS/Stat User's Guide. A list of all SAS/Stat procedures will come up. 

Click on the procedure that you wish to see. Each procedure has an introduction, a syntax guide, information on statistical algorithms, and examples using SAS code.The SAS help tab for the ANOVA procedure is shown below. All help is clickable.

You can also get help by going to the SAS support web page: http://support.sas.com. Click on Samples & SAS Notes where you can search for help using keywords. 

FASTats is a useful page that gives information on particular statistical topics, listed alphabetically. The url for this page is http://support.sas.com/kb/30/333.html

Getting Datasets into SAS
Before you can get started with SAS, you will first need either to read in some raw data, or open an existing SAS dataset.  

Reading in raw data in free format
Here is an excerpt of a raw data file that has each value separated by one or more blanks (free format).  The name of the raw data file is class.dat. Missing values are indicated by a period (.). Multiple missing values are indicated by inserting one or more blanks between periods for contiguous missing values.

Warren F 29 68 139

Kalbfleisch F 35 64 120

Pierce M . . 112

Walker F 22 56 133

Rogers M 45 68 145

Baldwin M 47 72 128

Mims F 48 67 152

Lambini F 36 . 120

Gossert M . 73 139

The SAS data step to read this type of raw data is shown below. The data statement names the data set to be created. The data set name can be no more than 32 characters long and can contain only letters, numbers and underscores (no blanks). The infile statement indicates the raw data file to be read. The input statement lists the variables to be read in the order in which they appear in the raw data file. No variables can be skipped at the beginning of the variable list, but you may stop reading variables before reaching the end of the list.  
data class;


infile "class.dat";


input lname $ sex $ age height sbp;

run;

Note that character variables are followed by a $. The default is numeric variables, which do not have a $ after them. The SAS log based on these commands is shown below, be sure to check the log to verify that the dataset was correctly created. The dataset name is WORK.CLASS. It is in the default WORK library, and is temporary.
1    data class;

2        infile "class.dat";

3        input lname $ sex $ age height sbp;

4    run;

NOTE: The infile "class.dat" is:

      Filename=C:\Users\kwelch\Desktop\b512\class.dat,

      RECFM=V,LRECL=256,File Size (bytes)=286,

      Last Modified=05Oct1998:00:44:32,

 NOTE: 14 records were read from the infile "class.dat".

      The minimum record length was 16.

      The maximum record length was 23.

NOTE: The data set WORK.CLASS has 14 observations and 5 variables.

SAS automatically looks for the raw data file (class.dat) in the current directory. You could also specify the exact path to this file in your infile statement, as shown below:
infile "c:\users\kwelch\desktop\labdata\class.dat";
To check the data, navigate to the Explorer Window, double-click on the Libraries Icon, 

Open the Work library: 

Double-click on the Class dataset. It will open in the Viewtable window. To close this view, click on the small X, not. the Red X, which will close all of SAS. 

You can have the dataset open in the Viewtable window for most SAS procedures, however you cannot sort or modify this dataset unless you close the viewtable window first.
Getting descriptive statistics
You can also get descriptive statistics for the dataset by using the commands shown below:

proc means data=class;

run;
                                The MEANS Procedure

         Variable     N            Mean         Std Dev         Minimum         Maximum

         ------------------------------------------------------------------------------

         age         12      36.2500000       7.8291414      22.0000000      48.0000000

         height      12      67.3333333       4.8116021      56.0000000      73.0000000

         sbp         14     133.8571429      11.0930133     112.0000000     152.0000000

         ------------------------------------------------------------------------------
Reading in raw data in column format
To read data that are lined up in columns, the input statement is set up by listing each variable followed by the column-range. Character variables are followed by a $, and then the column-range.  

Here is an example of a command file to read in raw data from marflt.dat. Proc print is used to print out the first 10 cases of the marflt data set.

data flights;


infile "marflt.dat" ;


input flight 1-3 depart $ 15-17 dest $ 18-20 boarded 34-36;

run;

proc print data=flights(obs=10);

run;
The output from these commands is shown below:
                       Obs    flight    depart   dest    boarded

                1      182      LGA      YYZ        104

                2      114      LGA      LAX        172

                3      202      LGA      ORD        151

                4      219      LGA      LON        198

                5      439      LGA      LAX        167

                6      387      LGA      CPH        152

                7      290      LGA      WAS         96

                8      523      LGA      ORD        177

                9      982      LGA      DFW         49

               10      622      LGA      FRA        207

Temporary SAS datasets

The datasets we have created so far have been temporary. Temporary SAS datasets are saved in the WORK library, and can go by a one-level name, e.g., Class, or by a two-level name, e.g., WORK.Class (to indicate that they are in the Work library). Temporary datasets will be lost at the end of the current session and must be re-created each time SAS is run.
Create a permanent SAS dataset

To create a permanent SAS dataset, you first need to submit a Libname statement to tell SAS where to save the data. A SAS library is an alias for a folder located in Windows. The folder must already exist before it can be referenced in a libname statement. The libname that you use must be 8 characters or less, and cannot start with a number (it must start with either a letter or underscore). 

libname mylib "C:\Users\kwelch\Desktop\labdata";
Check the log to be sure the libname was correctly specified. If you see a note that the library was successfully assigned, you are good to go. If you see an error message, check to be sure you have the correct path to the folder specified and that the folder already exists. If you have an error, correct it and resubmit your commands.

     libname mylib "C:\Users\kwelch\Desktop\labdata";

NOTE: Libref MYLIB was successfully assigned as follows:

      Engine:        V9

      Physical Name: C:\Users\kwelch\Desktop\labdata

Now you can create your permanent SAS dataset using commands like those below. Be sure you give the dataset a two-level name, where the first part of the name is the library (called a “libref” by SAS), and the second part of the name is the actual dataset name (called a “member” by SAS). Use a period between the two parts of the dataset name. The general form of a two-level dataset name is libref.datasetname.
data mylib.flights;


infile "marflt.dat" ;


input flight 1-3 depart $ 15-17 dest $ 18-20 boarded 34-36;

run;

proc means data=mylib.flights;

run;

Check your folder in windows to see the SAS dataset, which will be called flights.sas7bdat. The file extension (.sas7bdat) is not used in the SAS commands; it is automatically appended to the dataset name by SAS. 
Check the Log to be sure the dataset was correctly created. 

9    data mylib.flights;

10       infile "marflt.dat" ;

11       input flight 1-3 depart $ 15-17 dest $ 18-20 boarded 34-36;

12   run;

NOTE: The infile "marflt.dat" is:

      Filename=C:\Users\kwelch\Desktop\labdata\marflt.dat,

      RECFM=V,LRECL=256,File Size (bytes)=31751,

      Last Modified=21Dec1993:07:26:46,

      Create Time=21Dec1993:07:26:46

NOTE: Invalid data for boarded in line 420 34-36.

RULE:     ----+----1----+----2----+----3----+----4----+----5----+----6----+----7----+----8----+-

420       87203219013:02LGALAX2,475 283 5321-3  9  0210210 48

flight=872 depart=LGA dest=LAX boarded=. _ERROR_=1 _N_=420

NOTE: Invalid data for boarded in line 548 34-36.

548       92103279017:11LGADFW1,383 282 25012ª 16  5 79180 48

flight=921 depart=LGA dest=DFW boarded=. _ERROR_=1 _N_=548

NOTE: 635 records were read from the infile "marflt.dat".

      The minimum record length was 48.

      The maximum record length was 48.

NOTE: The data set MYLIB.FLIGHTS has 635 observations and 4 variables.

NOTE: DATA statement used (Total process time):

      real time           0.04 seconds

      cpu time            0.00 seconds

We note that there were some problems with the raw data when reading in this file, but the dataset was created with 635 observations and 4 variables.
Use permanent SAS datasets that already exist
To use a permanent SAS dataset or datasets, you first need to submit a Libname statement pointing to the folder where the dataset(s) are stored. 

libname sasdata2 "C:\Users\kwelch\Desktop\sasdata2";

You should see a note in the SAS log that shows that the library was successfully assigned.

     libname sasdata2 "C:\Users\kwelch\Desktop\sasdata2";

NOTE: Libref SASDATA2 was successfully assigned as follows:

      Engine:        V9

      Physical Name: C:\Users\kwelch\Desktop\sasdata2
If you get an error message, check to be sure the folder already exists and that you have specified it correctly.

Once this library is defined, you can view the datasets in it by going to the Explorer window and clicking on Libraries, and then selecting sasdata2. 

If you are already in the Work library, click in the explorer window, and go up one level to view all the current libraries. Then click on sasdata2 to see all the datasets in that library. Double-click on any dataset to view its contents. Close each dataset when you're done viewing it.

Use an existing SAS dataset
Once you have defined a library, you can use any dataset in that library. You just need to refer to it using a two-level name. The first part of the name is the library, followed by a dot and the dataset name. For example, you can refer to the Employee dataset by using its two-level name: sasdata2.employee. 

proc contents data=sasdata2.employee;

run;

The output from this procedure is shown below.
                                       The SAS System          

                                     The CONTENTS Procedure

    Data Set Name        SASDATA2.EMPLOYEE                        Observations          474

    Member Type          DATA                                     Variables             10

    Engine               V9                                       Indexes               0

    Created              Tuesday, January 13, 2009 12:44:50 PM    Observation Length    73

    Last Modified        Tuesday, January 13, 2009 09:23:54 AM    Deleted Observations  0

    Protection                                                    Compressed            NO

    Data Set Type                                                 Sorted                NO

    Label                Written by SAS

    Data Representation  WINDOWS_32

    Encoding             Default

                               Engine/Host Dependent Information

         Data Set Page Size          4096

         Number of Data Set Pages    10

         First Data Page             1

         Max Obs per Page            55

         Obs in First Data Page      25

         Number of Data Set Repairs  0

         Filename                    C:\Users\kwelch\Desktop\sasdata2\employee.sas7bdat

         Release Created             9.0000M0

         Host Created                WIN

                          Alphabetic List of Variables and Attributes

                  #    Variable    Type    Len    Label

                  3    bdate       Num       8    Date of Birth

                  4    educ        Num       8    Educational Level (years)

                  2    gender      Char      1    Gender

                  1    id          Num       8    Employee Code

                  5    jobcat      Num       8    Employment Category

                  8    jobtime     Num       8    Months since Hire

                 10    minority    Num       8    Minority Classification

                  9    prevexp     Num       8    Previous Experience (months)

                  6    salary      Num       8    Current Salary

                  7    salbegin    Num       8    Beginning Salary
Simple descriptive statistics can be obtained by using Proc Means:

proc means data=sasdata2.employee;

run;
                                      The MEANS Procedure

Variable    Label                             N            Mean         Std Dev         Minimum

-----------------------------------------------------------------------------------------------

id          Employee Code                   474     237.5000000     136.9762753       1.0000000

bdate       Date of Birth                   473        -1179.56         4302.33       -11282.00

educ        Educational Level (years)       474      13.4915612       2.8848464       8.0000000

jobcat      Employment Category             474       1.4113924       0.7732014       1.0000000

salary      Current Salary                  474        34419.57        17075.66        15750.00

salbegin    Beginning Salary                474        17016.09         7870.64         9000.00

jobtime     Months since Hire               474      81.1097046      10.0609449      63.0000000

prevexp     Previous Experience (months)    474      95.8607595     104.5862361               0

minority    Minority Classification         474       0.2194093       0.4142836               0

-----------------------------------------------------------------------------------------------

                    Variable    Label                                Maximum

                    --------------------------------------------------------

                    id          Employee Code                    474.0000000

                    bdate       Date of Birth                        4058.00

                    educ        Educational Level (years)         21.0000000

                    jobcat      Employment Category                3.0000000

                    salary      Current Salary                     135000.00

                    salbegin    Beginning Salary                    79980.00

                    jobtime     Months since Hire                 98.0000000

                    prevexp     Previous Experience (months)     476.0000000

                    minority    Minority Classification            1.0000000

                    --------------------------------------------------------                    
Frequencies for variables can be obtained using Proc Freq:

proc freq data=sasdata2.employee;

  tables gender jobcat;

run;

                                            Gender

                                                     Cumulative    Cumulative

                  gender    Frequency     Percent     Frequency      Percent

                  -----------------------------------------------------------

                  f              216       45.57           216        45.57

                  m              258       54.43           474       100.00

                                      Employment Category
                                                     Cumulative    Cumulative

                  jobcat    Frequency     Percent     Frequency      Percent

                  -----------------------------------------------------------

                       1         363       76.58           363        76.58

                       2          27        5.70           390        82.28

                       3          84       17.72           474       100.00
Formatting values to make output look nice
You can set up user-defined formats to display the output from your procedures so they look nicer. First, use Proc Format to define your formats. Note that there are two types of formats illustrated below. The first format is called $genderfmt and it uses a $ to indicate that it is a format that will be applied to a character variable. The second format is called jobcatfmt. It is for a numeric variable, so it doesn’t have the $ preceding the format name. Format names for character variables can be up to 31 characters long, and format names for numeric variables can be up to 32 characters long. Format names cannot end with a number.
proc format;

  value $genderfmt "f"="Female"

                   "m"="Male";

  value jobcatfmt 1="Clerical"

                  2="Custodial"




   3="Managerial";

run;

Check the log to be sure the formats were output correctly.

22   proc format;

23     value $genderfmt "f"="Female"

24                      "m"="Male";

NOTE: Format $GENDERFMT has been output.

25     value jobcatfmt 1="Clerical"

26                     2="Custodial"

27                     3="Managerial";

NOTE: Format JOBCATFMT has been output.

28   run;

Now that the formats have been created, they can be applied to any variable or variables in any dataset that you wish by using a format statement as part of the commands. For the format statement, just list the variable(s) following each variable or set of variables by the format name that applies to them. Be sure to use a period at the end of each format name in the format statement.
proc freq data=sasdata2.employee;

  tables gender jobcat;

  format gender $genderfmt. jobcat jobcatfmt.;

run;
The output from these commands is shown below.
                                            Gender

                                                     Cumulative    Cumulative

                  gender    Frequency     Percent     Frequency      Percent

                  -----------------------------------------------------------

                  Female         216       45.57           216        45.57

                  Male           258       54.43           474       100.00

                                      Employment Category

                                                       Cumulative    Cumulative

                    jobcat    Frequency     Percent     Frequency      Percent

                ---------------------------------------------------------------

                Clerical           363       76.58           363        76.58

                Custodial           27        5.70           390        82.28

                Managerial          84       17.72           474       100.00
Selecting cases for analysis

You can select cases for an analysis using the Where statement in your commands:

proc print data=sasdata2.employee;

  where jobcat=1;
run;

                                        The SAS System                                        

 Obs   id   gender    bdate   educ   jobcat   salary   salbegin   jobtime   prevexp   minority

   2    2     m        -588    16       1      40200     18750       98        36         0

   3    3     f      -11116    12       1      21450     12000       98       381         0

   4    4     f       -4644     8       1      21900     13200       98       190         0

   5    5     m       -1787    15       1      45000     21000       98       138         0

   6    6     m        -497    15       1      32100     13500       98        67         0

   7    7     m       -1345    15       1      36000     18750       98       114         0

   8    8     f        2317    12       1      21900      9750       98         0         0

   9    9     f       -5091    15       1      27900     12750       98       115         0

  10   10     f       -5070    12       1      24000     13500       98       244         0

  11   11     f       -3615    16       1      30300     16500       98       143         0

  12   12     m        2202     8       1      28350     12000       98        26         1

  13   13     m         198    15       1      27750     14250       98        34         1
Notice that some of the values of bdate are negative, because dates are stored as the number of days from January 1, 1960, with dates prior to this date having negative values. We will see how to make dates look nice later, here we use a SAS date format, and the dollar12. format plus our user-defined formats to make the data display look nicer:

proc print data=sasdata2.employee;

  where jobcat=1;

  format bdate mmddyy10. salary salbegin dollar12. gender $genderfmt.  

         jobcat jobcatfmt.;
run;
                                        The SAS System                                       

   Obs id gender      bdate educ  jobcat        salary     salbegin jobtime prevexp minority

     1  2 Male   05/23/1958  16  Clerical      $40,200      $18,750    98      36       0

     2  3 Female 07/26/1929  12  Clerical      $21,450      $12,000    98     381       0

     3  4 Female 04/15/1947   8  Clerical      $21,900      $13,200    98     190       0

     4  5 Male   02/09/1955  15  Clerical      $45,000      $21,000    98     138       0

     5  6 Male   08/22/1958  15  Clerical      $32,100      $13,500    98      67       0

     6  7 Male   04/26/1956  15  Clerical      $36,000      $18,750    98     114       0

     7  8 Female 05/06/1966  12  Clerical      $21,900       $9,750    98       0       0

     8  9 Female 01/23/1946  15  Clerical      $27,900      $12,750    98     115       0

     9 10 Female 02/13/1946  12  Clerical      $24,000      $13,500    98     244       0
Comments in a SAS program
There are two types of comments in a SAS program, both of which will appear in green in the Enhanced Editor.  You can start a comment with an asterisk (*) to comment out a single SAS statement. A semicolon (;) is required to terminate the comment. 
*This is an example of a comment;

**** This is also a valid comment ****;

You can use /* and */ to insert a comment anywhere in your SAS program. This type of comment can be used to comment out whole blocks of code.

/*This is an example of a comment*/

/************************************************

  This is also a valid comment

*************************************************/

Create and modify variables
The SAS Data Step is a powerful and flexible programming tool that is used to create a new SAS dataset.
The Data Step allows you to assign a particular value to all cases, or to a subset of cases; to transform a variable by using a mathematical function, such as the log function; or to create a sum, average, or other summary statistic based on the values of several existing variables within an observation.

NB:  A Data Step is required to create any new variables or modify existing variables in SAS. Unlike Stata and SPSS, you cannot simply create a new variable or modify an existing variable in “open” SAS code. You need to create a new dataset by using a Data Step whenever you want to create or modify variables. A single Data Step can be used to create an unlimited number of new variables. 

We will illustrate creating new variables using the employee dataset.
The Data Step starts with the Data statement and ends with Run. Each time you make any changes to the Data Step commands, you must highlight and re-submit the entire block of code, starting with "data" and ending with "run". This will re-create your dataset by over-writing the previous version. 

data sasdata2.employee2;

  set sasdata2.employee;

    /* put commands to create new variables here*/

    /* be sure they go BEFORE the run statement*/

run;
The example below illustrates creating a number of new variables in our new dataset. 
data sasdata2.employee2;

  set sasdata2.employee;

  currentyear=2005;

  alpha ="A"; 

  sept11 = "11SEP2001"D; 

  format Sept11 mmddyy10.;

  saldiff = salary - salbegin;

  if (salary > 50000) then salcat = "A" ;

  format bdate mmddyy10. salary salbegin dollar12.;

if gender="f" then female=1;

if gender="m" then female=0;

if jobcat not=. then do;

   jobdum1 = (jobcat=1);

   jobdum2 = (jobcat=2);

   jobdum3 = (jobcat=3);

end;

nmiss = nmiss(of educ--salbegin);

salmean = mean(salary, salbegin);

run;
Run these commands and then browse your new dataset to see the results.
Examples of Functions and Operators in SAS

The following list contains some of the more common SAS functions and operators:

Arithmetic Operators:

+
Addition

-
Subtraction

*
Multiplication

/
Division

**
Exponentiation

Arithmetic Functions:

ABS

Absolute value

ROUND(arg,unit) 
Rounds argument










to the nearest unit 

INT

Truncate


MOD


Modulus 










(remainder)

SQRT

Square root


EXP


Exponential

LOG10
Log base 10


LOG


Natural log

SIN

Sine



COS


Cosine

Statistical Functions (Arguments can be numeric values or variables):

SUM(Arg1, Arg2,…,ArgN)


Sum of non-missing arguments

MEAN(Arg1, Arg2,…,ArgN)


Mean of non-missing arguments

STD(Arg1, Arg2,…,ArgN)


Standard deviation of non-missing 
arguments

VAR(Arg1, Arg2,…,ArgN)


Variance of non-missing arguments

CV(Arg1, Arg2,…,ArgN)
Coefficient of variation of non-missing arguments

MIN(Arg1, Arg2,…,ArgN)


Minimum of non-missing arguments

MAX(Arg1, Arg2,…,ArgN)


Maximum of non-missing arguments

Missing Values Functions:
MISSING(Arg)
= 1 if the value of Arg is missing         
= 0 if not missing

NMISS(Var1, Var2,…,VarN)
Number of missing values across variables within a case

N(Var1, Var2,…,VarN)
Number of non-missing values across variables within a case

Across-case Functions:
LAG(Var)




Value from previous case

LAGn(Var)




Value from nth previous case

Date and Time Functions:

Datepart(datetimevalue)
Extracts date portion from a datetime value

Month(datevalue)



Extracts month from a date value

Day(datevalue)



Extracts day from a date value

Year(datevalue)



Extracts year form a date value

Intck(‘interval’,datestart,dateend)
Finds the number of completed intervals between two dates

Other Functions:
RANUNI(Seed)
Uniform pseudo-random no. defined on the interval (0,1)

RANNOR(Seed)



Std. Normal pseudo-random no.

PROBNORM(x)



Prob. a std. normal  is <= x

PROBIT(p) 




pth quantile from std. normal dist.

Numeric vs. Character Variables

There are only two types of variable in SAS: numeric and character. Numeric variables are the default type and are used for numeric and date values. 

Character variables can have alpha-numeric values, which may be any combination of letters, numbers, or other characters. The length of a character variable can be up to 32767 characters. Values of character variables are case-sensitive. For example, the value “Ann Arbor” is different than the value “ANN ARBOR”. 

Generating Variables Containing Constants

In the example below we create a new numeric variable named “currentyear”, which has a constant value of 2005 for all observations:

currentyear=2005;

The example below illustrates creating a new character variable named “alpha” which contains the letter “A” for all observations in the dataset. Note that the value must be enclosed either in single or double-quotes, because this is a character variable. 

alpha ="A"; 

Dates can be generated in a number of different ways. For example, we can use the mdy function to create a date value from a month, day, and year value, as shown below:

sept11 = mdy(9,11,2001); 

Or we can create a date by using a SAS date constant, as shown below:

sept11 = "11SEP2001"D; 

The D following the quoted date constant tells SAS that this is not a character variable, but a date value, which is stored as a numeric value.

format Sept11 mmddyy10.;

The format statement tells SAS to display the date as 09/11/2001, rather than as the number of days from January 1, 1960. 

Generating Variables Using Values from Other Variables

We can also generate new variables as a function of existing variables. 

saldiff = salary – salbegin;

New variables can be labeled with a descriptive label up to 40 characters long:

label saldiff = "Current Salary – Beginning Salary";

We can use the mdy function to create a new date value, based on the values of three variables, in this example the variables were called “Month”, “Day”, and “Year”, although they could have different names:

date = mdy(month,day,year);

Values of the date variable would vary from observation to observation, because the mdy() function is using different values of variables to create date. Remember to use a Format statement to format the new variable DATE so it will look like a date.

format date mmddyy10.;

Generating Variables Conditionally Based on Values of Other Variables

You can also create new variables in SAS conditional on the values of other variables. For example, if we wanted to create a new character variable, SALCAT, that contains salary categories “A”, “B”, and “C” we could use the following commands. 

if (salary >= 0 and salary <= 25000) then salcat = "C"; 

if (salary > 25000 & salary <= 50000) then salcat = "B";

if (salary > 50000) then salcat = "A";
Note the use of an If…Then statement to identify the condition that a given case in the data set must meet for the new variable to be given a value of “A”. In general, these types of conditional commands have the form:

if (condition) then varname = value;

where the condition can be specified using a logical operator or a mnemonic (e.g., =  (eq), & (and), | (or), ~= (not=, ne), > (gt), >= (ge) < (lt) <= (le)). The parentheses are not necessary to specify a condition in SAS, but can be used to clarify a statement or to group parts of a statement. A semicolon is required at the end of the statement. For example, if one wants to create a variable that identifies employees who are managers but have relatively low salaries, one could use a statement like

if (salary < 50000 & jobcat = 3) then manlowsal = 1;

This will create a new character variable equal to 1 whenever an employee meets the specified conditions on the two variables, salary and jobcat. However, this variable may be incorrectly coded, due to the presence of missing values, as discussed in the note below. 

Note on missing values when conditionally computing new variables in SAS:

SAS considers missing values for numeric variables to be smaller than the smallest possible numeric value in a data set. Therefore, in the salary condition above, if an employee had missing data on the salary variable, that employee would be coded into category 1 on the new MANLOWSAL variable. A safer version of this conditional command would look like this:

if (salary not=. & salary < 50000 & jobcat = 3) then manlowsal = 1; 

The condition now emphasizes that salary must be less than $50,000 and not equal to a missing value. 

The following statements could be used to set up a variable with a value of 1 or 0 on the new variable MANLOWSAL. Note that the use of ‘else’ will put all values, including missing values on either variable, into the 0 category (every other value, including missing, is captured by the ‘else’ condition). The final If statement will put anyone with a missing value on either of these variables into the missing value of MANLOWSAL, which is

 . for a numeric variable. 

if (salary not=. & salary < 50000 & jobcat = 3) then manlowsal =1; 

else manlowsal = 0;

if salary = . or jobcat=. then manlowsal= . ;

Another way this could be done would be to use a Do Loop before creating the variable, as shown below. If you use a do; statement, you must have an end; statement to close the do loop. In the example below, the entire block of code will only be executed if salary is not missing and jobcat is not missing.

if salary not=. and jobcat not=. then do;

  if (salary < 50000 & jobcat = 3) then manlowsal = 1; 

  else manlowsal = 0;

end;

Generating Dummy Variables

Statistical analyses often require dummy variables, which are also known as indicator variables. Dummy variables take on a value of 1 for certain cases, and 0 for all other cases. A common example is the creation of a dummy variable to recode, where the value of 1 might identify females, and 0 males. 

if gender="f" then female=1;

if gender="m" then female=0;

If you have a variable with 3 or more categories, you can create a dummy variable for each category, and later in a regression analysis, you would usually choose to include one less dummy variable than there are categories in your model. 
if jobcat not=. then do;

   jobdum1 = (jobcat=1);

   jobdum2 = (jobcat=2);

   jobdum3 = (jobcat=3);

end;

Using Statistical Functions

You can also use SAS to determine how many missing values are present in a list of variables within an observation, as shown in the example below:

nmiss = nmiss(of educ--salbegin);

The double dashes (--) indicate a variable list (with variables given in dataset order). Be sure to use “of” when using a variable list like this.

The converse operation is to determine the number of non-missing values there are in a list of variables, 

npresent = n(of educ--salbegin);

Another common operation is to calculate the sum or the mean of the values for several variables and store the results in a new variable. For example, to calculate a new variable, salmean, representing the average of the current and beginning salary, use the following command. Note that you can use a list of variables separated by commas, without including “of” before the list.

salmean = mean(salary, salbegin);

All missing values for the variables listed will be ignored when computing the mean in this way. The min( ), max( ), and std( ) functions work in a similar way. 

Permanent Formats

Previously we saw how to create and use temporary formats. Another way to assign formats so that they will apply to the variables any time they are used is to use a format statement in a data step. This method will assign the formats to the variables at any time they are used in the future for any procedure. However, it has a disadvantage in that SAS will give you an error message if the formats for a particular variable are not found.

The commands below show how to set up and use permanent SAS formats. First, create the formats, and put them in a formats catalog in a library (in this case we chose to put them in SASDATA2).   
proc format lib=sasdata2;

  value jobcatfmt 1="Clerical" 2="Custodial" 3="Manager";

  value $genderfmt "f" = "Female"

                    "m" = "Male";

  value minfmt  1="Yes"

                0="No";

run;
Now, tell SAS where to locate the formats.

options fmtsearch=(work sasdata2);
Next, run Proc Datasets to apply the formats to the variables in the dataset. In this case we are applying the formats to the sasdata2.employee2 dataset.
proc datasets lib=sasdata2;

  modify employee2;

  format jobcat jobcatfmt. gender $genderfmt. minority minfmt.;

run;
This produces output in the SAS log that shows how the employee dataset in the sasdata2 library has been modified.
104  proc datasets lib=sasdata2;

                                           Directory

                        Libref         SASDATA2

                        Engine         V9

                        Physical Name  C:\Users\kwelch\Desktop\sasdata2

                        Filename       C:\Users\kwelch\Desktop\sasdata2

                                      Member       File

             #  Name                  Type         Size  Last Modified

             1  AFIFI                 DATA        50176  08Dec10:11:10:13

             2  AUTISM_DEMOG          DATA        25600  03Mar08:10:42:22

             3  AUTISM_SOCIALIZATION  DATA        41984  24Jul09:06:40:00

             4  BANK                  DATA        46080  13Jan08:19:17:04

             5  BASEBALL              DATA        82944  20Jun02:06:12:32

             6  BUSINESS              DATA        17408  20Aug06:03:05:04

             7  BUSINESS2             DATA        17408  24Sep10:05:20:10

             8  CARS                  DATA        33792  22Aug06:00:03:42

             9  EMPLOYEE              DATA        41984  10Jan11:13:58:24

            10  EMPLOYEE2             DATA        13312  10Jan11:14:10:12

            11  FITNESS               DATA         9216  06Mar06:09:04:44

            12  FORMATS               CATALOG     17408  10Jan11:14:11:31

            13  IRIS                  DATA        13312  20Jun02:06:12:32

            14  TECUMSEH              DATA      1147904  01Jun05:23:00:04

            15  WAVE1                 DATA       578560  20Jun07:11:11:00

            16  WAVE2                 DATA       492544  20Jun07:11:11:00

            17  WAVE3                 DATA       455680  20Jun07:11:11:00

105    modify employee2;

106    format jobcat jobcats. gender $sexfmt. minority minfmt.;

107  run;

NOTE: MODIFY was successful for SASDATA2.EMPLOYEE2.DATA.

108  quit;
You can see that the formats have been applied by checking Proc Contents:

proc contents data=sasdata2.employee2;

run;
The list of variables from Proc contents is shown below:

                   Alphabetic List of Variables and Attributes

       #    Variable       Type    Len    Format         Label

      12    alpha          Char      1

       3    bdate          Num       8    MMDDYY10.      Date of Birth

      11    currentyear    Num       8

       4    educ           Num       8                   Educational Level (years)

      17    female         Num       8

       2    gender         Char      1    $GENDERFMT.    Gender

       1    id             Num       8                   Employee Code

       5    jobcat         Num       8    JOBCATFMT.     Employment Category

      18    jobdum1        Num       8

      19    jobdum2        Num       8

      20    jobdum3        Num       8

       8    jobtime        Num       8                   Months since Hire

      16    manlowsal      Num       8

      10    minority       Num       8    MINFMT.        Minority Classification

      21    nmiss          Num       8

       9    prevexp        Num       8                   Previous Experience (months)

       6    salary         Num       8    DOLLAR12.      Current Salary

       7    salbegin       Num       8    DOLLAR12.      Beginning Salary

      15    salcat         Num       8
To use this dataset with the formats later, you can use SAS code similar to that shown below:
options nofmterr;

libname sasdata2 "C:\Users\kwelch\Desktop\sasdata2";

options fmtsearch=(work sasdata2);
proc means data=sasdata2.employee2;

  class jobcat;
  var saldiff;
run;
Translation:

options nofmterr; 

  This statement tells SAS not to produce an error if there is a problem with the formats. You want to have this in as insurance, in case anything goes wrong with your formats.
libname sasdata2 "C:\Users\kwelch\Desktop\sasdata2";

  This statement defines the library where the formats catalog and the SAS dataset are located.

options fmtsearch=(work sasdata2);
  This statement tells SAS where to search for formats to be used with the datasets that are used in the session.
The output from these commands is shown below:

                                  Analysis Variable : saldiff

    Employment      N

    Category      Obs      N            Mean         Std Dev         Minimum         Maximum

    ----------------------------------------------------------------------------------------

    Clerical      363    363        13742.49         5973.77         5550.00        64250.00

    Custodial      27     27        15861.11         2415.31         9300.00        21750.00

    Manager        84     84        33719.94        13424.34        14250.00        76240.00

    ----------------------------------------------------------------------------------------
Sorting data
Often you may want to stratify your data, getting a separate analysis for different subgroups. This can be done by first sorting the data “by” a variable or variables and then using SAS Procs with a “By” statement, as shown below: You can use a “By” statement with most SAS procedures.

proc sort data=sasdata2.employee2;

  by gender;

run;
proc means data=sasdata2.employee2;

  by gender;

run;
---------------------------------------- Gender=Female -----------------------------------------

                                      The MEANS Procedure

 Variable     Label                                N          Mean       Std Dev       Minimum

 ---------------------------------------------------------------------------------------------

 id           Employee Code                      216   249.3842593   131.1307373     3.0000000

 bdate        Date of Birth                      216      -1420.00       4992.10     -11282.00

 educ         Educational Level (years)          216    12.3703704     2.3191522     8.0000000

 jobcat       Employment Category                216     1.0925926     0.4212283     1.0000000

 salary       Current Salary                     216      26031.92       7558.02      15750.00

 salbegin     Beginning Salary                   216      13091.97       2935.60       9000.00

 jobtime      Months since Hire                  216    80.3796296     9.6763606    63.0000000

 prevexp      Previous Experience (months)       216    77.0370370    95.0120605             0

 minority     Minority Classification            216     0.1851852     0.3893500             0

 currentyear                                     216       2005.00             0       2005.00

 sept11                                          216      15229.00             0      15229.00

 saldiff      Current Salary – Beginning Salary  216      12939.95       5661.08       5550.00

 salcat                                          216     1.4537037     0.5521082     1.0000000

 manlowsal                                       216     0.0277778     0.1647173             0

 female                                          216     1.0000000             0     1.0000000

 jobdum1                                         216     0.9537037     0.2106141             0

 jobdum2                                         216             0             0             0

 jobdum3                                         216     0.0462963     0.2106141             0

 nmiss                                           216             0             0             0

 salmean                                         216      19561.94       4985.85      12975.00

 --------------------------------------------------------------------------------------------

----------------------------------------- Gender=Male ------------------------------------------

 Variable     Label                                N          Mean       Std Dev       Minimum

 ---------------------------------------------------------------------------------------------

 id           Employee Code                      258   227.5503876   141.1684405     1.0000000

 bdate        Date of Birth                      257  -977.4785992       3620.54     -11200.00

 educ         Educational Level (years)          258    14.4302326     2.9793349     8.0000000

 jobcat       Employment Category                258     1.6782946     0.8915916     1.0000000

 salary       Current Salary                     258      41441.78      19499.21      19650.00

 salbegin     Beginning Salary                   258      20301.40       9111.78       9000.00

 jobtime      Months since Hire                  258    81.7209302    10.3510196    63.0000000

 prevexp      Previous Experience (months)       258   111.6201550   109.6922965     3.0000000

 minority     Minority Classification            258     0.2480620     0.4327274             0

 currentyear                                     258       2005.00             0       2005.00

 sept11                                          258      15229.00             0      15229.00

 saldiff      Current Salary – Beginning Salary  258      21140.39      12557.06       5550.00

 salcat                                          258     2.1782946     0.5430797     1.0000000

 manlowsal                                       258     0.0465116     0.2109997             0

 female                                          258             0             0             0

 jobdum1                                         258     0.6085271     0.4890284             0

 jobdum2                                         258     0.1046512     0.3066984             0

 jobdum3                                         258     0.2868217     0.4531566             0

 nmiss                                           258             0             0             0

 salmean                                         258      30871.59      13863.69      16125.00

 ---------------------------------------------------------------------------------------------

By default SAS sorts observations in Ascending order, from smallest to largest. The example below sorts the dataset with salary= . (i.e. missing salary) coming first, and then the values of salary. It is important to know that the dataset is physically replaced after being sorted, and that the sorting will be in place until the dataset is either re-created or re-sorted. Be sure you don’t have a dataset open in the Viewtable window when you’re trying to sort it!
proc sort data=sasdata2.employee2;

  by salary;

run;
Check the log to be sure the sorting happened correctly.

55   proc sort data=sasdata2.employee2;

56     by salary;

57   run;

NOTE: There were 474 observations read from the data set SASDATA2.EMPLOYEE2.

NOTE: The data set SASDATA2.EMPLOYEE2 has 474 observations and 22 variables.

You can see that the dataset is sorted by using Proc Contents. The sort information will be at the bottom of the output from Proc Contents.
proc contents data=sasdata2.employee2;

run;

                                       Sort Information

                                     Sortedby       salary

                                     Validated      YES

                                     Character Set  ANSI

To sort by descending order, using the descending option:

proc sort data=sasdata2.employee2;

  by descending salary;

run;

Sort by more than one variable
When sorting by more than one variable, SAS sorts based on values of the first variable, and then it sorts based on the second variable, and so on.  By default, SAS will use Ascending order for each variable, unless otherwise specified. 

proc sort data=sasdata2.employee2;

  by gender salary;

run;

Double-click here to change the current directory.

